**************************************************************
* ELECTRICITY SPILOVERS PROJECT
*  ROBUSTNESS 
**************************************************************
set more off
frame copy data_main robust
frame change robust
keep if post==1


********************************************************************************
*TABLE A.1 - ELECTRICITY INTENT-TO-TREAT EFFECTS (RICH FIXED EFFECTS)
********************************************************************************
gen dow=dow(date)
gen month=month(date)

eststo clear
*Column 1: Full Diff + weather + date + DoWXHODXMonth FEs
	eststo: reghdfe euse ws temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1, ///
			absorb(date i.dow#i.hour#i.month) vce(cluster HHID)
	sum euse if e(sample) & ws==0
	scalar euse1 = r(mean) 
		estadd scalar m_cont = scalar(euse1)
		estadd local weath "Yes"
		estadd local rich "Yes"	
		estadd local date "Yes"
		estadd local hours "All"
		estadd local sam "05/2015-05/2016"		
	est store A
	
*Column 2: Full Diff + weather + date + HOD FE + pre-treatment use
	eststo: reghdfe euse ws elec_use_summer elec_use_annual elec_use_winter temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1, ///
			absorb(date i.dow#i.hour#i.month) vce(cluster HHID)
	sum euse if e(sample) & ws==0
	scalar euse2 = r(mean)
		estadd scalar m_cont = scalar(euse2)
		estadd local weath "Yes"
		estadd local rich "Yes"	
		estadd local date "Yes"
		estadd local hours "All"
		estadd local sam "05/2015-05/2016"	
	est store B
		
*Column 3: Summer 2015 Diff weather + date + HOD FE 
	eststo: reghdfe euse ws temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1 if ym<=667, ///
			absorb(date i.dow#i.hour#i.month) vce(cluster HHID)
	sum euse if e(sample) & ws==0
	scalar euse3 = r(mean)
		estadd scalar m_cont = scalar(euse3)
		estadd local weath "Yes"
		estadd local rich "Yes"	
		estadd local date "Yes"
		estadd local hours "All"
		estadd local sam "05/2015-08/2015"					
	est store C
	
*Column 4: Summer 2015 Diff + weather + date + HOD FE + pre-treatment use
	eststo: reghdfe euse ws elec_use_summer elec_use_annual elec_use_winter temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1 if ym<=667, ///
			absorb(date i.dow#i.hour#i.month) vce(cluster HHID)
	sum euse if e(sample) & ws==0
	scalar euse4 = r(mean)
		estadd scalar m_cont = scalar(euse4)
		estadd local weath "Yes"
		estadd local rich "Yes"	
		estadd local date "Yes"
		estadd local hours "All"
		estadd local sam "05/2015-08/2015"					
	est store D
		
*Column 5: Summer 2015 Diff weather + date + HOD FE - Peak Hours
	eststo: reghdfe euse ws temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1 if ym<=667 & (hour>=15 & hour<=20), ///
			absorb(date i.dow#i.hour#i.month) vce(cluster HHID)
	sum euse if e(sample) & ws==0
	scalar euse5 = r(mean)
		estadd scalar m_cont = scalar(euse5)
		estadd local weath "Yes"
		estadd local rich "Yes"	
		estadd local date "Yes"
		estadd local hours "Peak"
		estadd local sam "05/2015-08/2015"					
	est store E
	
*Column 6: Summer 2015 Diff + weather + date + HOD FE  + pre-treatment use- Peak Hours
	eststo: reghdfe euse ws elec_use_summer elec_use_annual elec_use_winter temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1 if  ym<=667 & (hour>=15 & hour<=20), ///
			absorb(date i.dow#i.hour#i.month) vce(cluster HHID)
	sum euse if e(sample) & ws==0
	scalar euse6 = r(mean)
		estadd scalar m_cont = scalar(euse6)
		estadd local weath "Yes"
		estadd local rich "Yes"	
		estadd local date "Yes"
		estadd local hours "Peak"
		estadd local sam "05/2015-08/2015"					
	est store F
	
esttab A B C D E F using $tables\eregs_richfe_$outputdate.tex, replace label ///
    booktabs b(a2) nonumber ///
	drop(temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1) ///
    starlevels(* 0.10 ** 0.05 *** 0.01) ///
	title(Electricity Intent-to-Treat Effects ///
	     (Dependent Variable: Electricity Use (kWh/hr)) \label{eregs1})  ///
	cells(b(fmt(3) star) se(fmt(3) par)) ///
	note(Notes: The table reports intent-to-treat results from an OLS ///
		 regression of hourly electricity use on assignment to the treatment. ///
		 Columns 1 and 2 include all observations from May 15, 2015 to May ///
		 31, 2016. Columns 3 and 4 restrict the sample to the summer of 2015 ///
		 (May 15 to August 30). Columns 5 and 6 further limit the sample to ///
		 include only peak demand hours (3 PM to 8 PM). Standard errors are ///
		 clustered at the household. *, **, *** denote significance at the ///
		 10\%, 5\%, and 1\%level.) ///
	scalars("m_cont Mean Control Group Use" "weath Weather Controls" /// 
		    "rich HODxDOWxMonth FE" "date Calendar Date FE" ///
			 "hours Hours" "sam Sample") ///
	 mlabels((1) (2) (3) (4) (5) (6)) collabels(none) 
	
   
********************************************************************************
*TABLE A.2 - ELECTRICITY INTENT-TO-TREAT EFFECTS (MONTHLY PRE-TREATMENT CONTROLS)
********************************************************************************
eststo clear

*Column 2: Full Diff + weather + date + HOD FE + pre-treatment use
	eststo: reghdfe euse ws mar14_use-feb15_use temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1, ///
			absorb(hour date) vce(cluster HHID)
	sum euse if e(sample) & ws==0
	scalar euse2 = r(mean)
		estadd scalar m_cont = scalar(euse2)
		estadd local weath "Yes"
		estadd local hod "Yes"	
		estadd local date "Yes"
		estadd local hours "All"
		estadd local sam "05/2015-05/2016"	
	est store A
		
	
*Column 4: Summer 2015 Diff + weather + date + HOD FE + pre-treatment use
	eststo: reghdfe euse ws mar14_use-feb15_use temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1 if ym<=667 & elec_use_summer!=., ///
			absorb(hour date) vce(cluster HHID)
	sum euse if e(sample) & ws==0
	scalar euse4 = r(mean)
		estadd scalar m_cont = scalar(euse4)
		estadd local weath "Yes"
		estadd local hod "Yes"	
		estadd local date "Yes"
		estadd local hours "All"
		estadd local sam "05/2015-08/2015"					
	est store B
	
*Column 6: Summer 2015 Diff + weather + date + HOD FE  + pre-treatment use- Peak Hours
	eststo: reghdfe euse ws mar14_use-feb15_use temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1 if elec_use_summer!=. & ym<=667 & (hour>=15 & hour<=20), ///
			absorb(hour date) vce(cluster HHID)
	sum euse if e(sample) & ws==0
	scalar euse6 = r(mean)
		estadd scalar m_cont = scalar(euse6)
		estadd local weath "Yes"
		estadd local hod "Yes"	
		estadd local date "Yes"
		estadd local hours "Peak"
		estadd local sam "05/2015-08/2015"					
	est store C
	

esttab A B C using $tables\eregs_premonth_$outputdate.tex, replace label ///
    booktabs b(a2) nonumber ///
	drop(temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1) ///
    starlevels(* 0.10 ** 0.05 *** 0.01) ///
	title(Electricity Intent-to-Treat Effects ///
	     (Dependent Variable: Electricity Use (kWh/hr)) \label{eregs1})  ///
	cells(b(fmt(3) star) se(fmt(3) par)) ///
	note(Notes: The table reports intent-to-treat results from an OLS ///
		 regression of hourly electricity use on assignment to the treatment. ///
		 Columns 1 and 2 include all observations from May 15, 2015 to May ///
		 31, 2016. Columns 3 and 4 restrict the sample to the summer of 2015 ///
		 (May 15 to August 30). Columns 5 and 6 further limit the sample to ///
		 include only peak demand hours (3 PM to 8 PM). Standard errors are ///
		 clustered at the household. *, **, *** denote significance at the ///
		 10\%, 5\%, and 1\%level.) ///
	scalars("m_cont Mean Control Group Use" "weath Weather Controls" /// 
		    "hod Hour of Day FE" "date Calendar Date FE" ///
			 "hours Hours" "sam Sample") ///
	 mlabels((1) (2) (3) (4) (5) (6)) collabels(none) 

	
	
********************************************************************************
*TABLE A.3: ELECTRICITY INTENT-TO-TREAT EFFECTS - LOG AND HYBERBOLIC SINE OF ELECTRICITY USE
********************************************************************************
gen leuse1=log(euse+1)
gen leuse2=asinh(euse)

************************************
*Log(euse+1)

 eststo clear
*Column 1: Full Diff + weather + date + HOD FE
	eststo: reghdfe leuse1 ws temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1, ///
			absorb(hour date) vce(cluster HHID)
	sum euse if e(sample) & ws==0
	scalar euse1 = r(mean) 
		estadd scalar m_cont = scalar(euse1)
		estadd local weath "Yes"
		estadd local hod "Yes"	
		estadd local date "Yes"
		estadd local hours "All"
		estadd local sam "05/2015-05/2016"		
	est store A
	
*Column 2: Full Diff + weather + date + HOD FE + pre-treatment use
	eststo: reghdfe leuse1 ws elec_use_summer elec_use_annual elec_use_winter temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1, ///
			absorb(hour date) vce(cluster HHID)
	sum euse if e(sample) & ws==0
	scalar euse2 = r(mean)
		estadd scalar m_cont = scalar(euse2)
		estadd local weath "Yes"
		estadd local hod "Yes"	
		estadd local date "Yes"
		estadd local hours "All"
		estadd local sam "05/2015-05/2016"	
	est store B
		
*Column 3: Summer 2015 Diff weather + date + HOD FE 
	eststo: reghdfe leuse1 ws temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1 if ym<=667, ///
			absorb(hour date) vce(cluster HHID)
	sum euse if e(sample) & ws==0
	scalar euse3 = r(mean)
		estadd scalar m_cont = scalar(euse3)
		estadd local weath "Yes"
		estadd local hod "Yes"	
		estadd local date "Yes"
		estadd local hours "All"
		estadd local sam "05/2015-08/2015"					
	est store C
	
*Column 4: Summer 2015 Diff + weather + date + HOD FE + pre-treatment use
	eststo: reghdfe leuse1 ws elec_use_summer elec_use_annual elec_use_winter temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1 if ym<=667, ///
			absorb(hour date) vce(cluster HHID)
	sum euse if e(sample) & ws==0
	scalar euse4 = r(mean)
		estadd scalar m_cont = scalar(euse4)
		estadd local weath "Yes"
		estadd local hod "Yes"	
		estadd local date "Yes"
		estadd local hours "All"
		estadd local sam "05/2015-08/2015"					
	est store D
		
*Column 5: Summer 2015 Diff weather + date + HOD FE - Peak Hours
	eststo: reghdfe leuse1 ws temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1 if ym<=667 & (hour>=15 & hour<=20), ///
			absorb(hour date) vce(cluster HHID)
	sum euse if e(sample) & ws==0
	scalar euse5 = r(mean)
		estadd scalar m_cont = scalar(euse5)
		estadd local weath "Yes"
		estadd local hod "Yes"	
		estadd local date "Yes"
		estadd local hours "Peak"
		estadd local sam "05/2015-08/2015"					
	est store E
	
*Column 6: Summer 2015 Diff + weather + date + HOD FE  + pre-treatment use- Peak Hours
	eststo: reghdfe leuse1 ws elec_use_summer elec_use_annual elec_use_winter temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1 if   ym<=667 & (hour>=15 & hour<=20), ///
			absorb(hour date) vce(cluster HHID)
	sum euse if e(sample) & ws==0
	scalar euse6 = r(mean)
		estadd scalar m_cont = scalar(euse6)
		estadd local weath "Yes"
		estadd local hod "Yes"	
		estadd local date "Yes"
		estadd local hours "Peak"
		estadd local sam "05/2015-08/2015"					
	est store F
	
	
esttab A B C D E F using $tables\eregs_log1_$outputdate.tex, replace label ///
    booktabs b(a2) nonumber ///
	drop(temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1) ///
    starlevels(* 0.10 ** 0.05 *** 0.01) ///
	title(Electricity Intent-to-Treat Effects ///
	     (Dependent Variable: Electricity Use (kWh/hr)) \label{eregs1})  ///
	cells(b(fmt(3) star) se(fmt(3) par)) ///
	note(Notes: The table reports intent-to-treat results from an OLS ///
		 regression of hourly electricity use on assignment to the treatment. ///
		 Columns 1 and 2 include all observations from May 15, 2015 to May ///
		 31, 2016. Columns 3 and 4 restrict the sample to the summer of 2015 ///
		 (May 15 to August 30). Columns 5 and 6 further limit the sample to ///
		 include only peak demand hours (3 PM to 8 PM). Standard errors are ///
		 clustered at the household. *, **, *** denote significance at the ///
		 10\%, 5\%, and 1\%level.) ///
	scalars("m_cont Mean Control Group Use" "weath Weather Controls" /// 
		    "hod Hour of Day FE" "date Calendar Date FE" ///
			 "hours Hours" "sam Sample") ///
	 mlabels((1) (2) (3) (4) (5) (6)) collabels(none) 
	 
	 
************************************
*sinh^-1(euse)	 
 eststo clear
*Column 1: Full Diff + weather + date + HOD FE
	eststo: reghdfe leuse2 ws temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1, ///
			absorb(hour date) vce(cluster HHID)
	sum euse if e(sample) & ws==0
	scalar euse1 = r(mean) 
		estadd scalar m_cont = scalar(euse1)
		estadd local weath "Yes"
		estadd local hod "Yes"	
		estadd local date "Yes"
		estadd local hours "All"
		estadd local sam "05/2015-05/2016"		
	est store A
	
*Column 2: Full Diff + weather + date + HOD FE + pre-treatment use
	eststo: reghdfe leuse2 ws elec_use_summer elec_use_annual elec_use_winter temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1, ///
			absorb(hour date) vce(cluster HHID)
	sum euse if e(sample) & ws==0
	scalar euse2 = r(mean)
		estadd scalar m_cont = scalar(euse2)
		estadd local weath "Yes"
		estadd local hod "Yes"	
		estadd local date "Yes"
		estadd local hours "All"
		estadd local sam "05/2015-05/2016"	
	est store B
		
*Column 3: Summer 2015 Diff weather + date + HOD FE 
	eststo: reghdfe leuse2 ws temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1 if ym<=667, ///
			absorb(hour date) vce(cluster HHID)
	sum euse if e(sample) & ws==0
	scalar euse3 = r(mean)
		estadd scalar m_cont = scalar(euse3)
		estadd local weath "Yes"
		estadd local hod "Yes"	
		estadd local date "Yes"
		estadd local hours "All"
		estadd local sam "05/2015-08/2015"					
	est store C
	
*Column 4: Summer 2015 Diff + weather + date + HOD FE + pre-treatment use
	eststo: reghdfe leuse2 ws elec_use_summer elec_use_annual elec_use_winter temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1 if ym<=667, ///
			absorb(hour date) vce(cluster HHID)
	sum euse if e(sample) & ws==0
	scalar euse4 = r(mean)
		estadd scalar m_cont = scalar(euse4)
		estadd local weath "Yes"
		estadd local hod "Yes"	
		estadd local date "Yes"
		estadd local hours "All"
		estadd local sam "05/2015-08/2015"					
	est store D
		
*Column 5: Summer 2015 Diff weather + date + HOD FE - Peak Hours
	eststo: reghdfe leuse2 ws temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1 if ym<=667 & (hour>=15 & hour<=20), ///
			absorb(hour date) vce(cluster HHID)
	sum euse if e(sample) & ws==0
	scalar euse5 = r(mean)
		estadd scalar m_cont = scalar(euse5)
		estadd local weath "Yes"
		estadd local hod "Yes"	
		estadd local date "Yes"
		estadd local hours "Peak"
		estadd local sam "05/2015-08/2015"					
	est store E
	
*Column 6: Summer 2015 Diff + weather + date + HOD FE  + pre-treatment use- Peak Hours
	eststo: reghdfe leuse2 ws elec_use_summer elec_use_annual elec_use_winter temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1 if ym<=667 & (hour>=15 & hour<=20), ///
			absorb(hour date) vce(cluster HHID)
	sum euse if e(sample) & ws==0
	scalar euse6 = r(mean)
		estadd scalar m_cont = scalar(euse6)
		estadd local weath "Yes"
		estadd local hod "Yes"	
		estadd local date "Yes"
		estadd local hours "Peak"
		estadd local sam "05/2015-08/2015"					
	est store F
	
esttab A B C D E F using $tables\eregs_log2_$outputdate.tex, replace label ///
    booktabs b(a2) nonumber ///
	drop(temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1) ///
    starlevels(* 0.10 ** 0.05 *** 0.01) ///
	title(Electricity Intent-to-Treat Effects ///
	     (Dependent Variable: Electricity Use (kWh/hr)) \label{eregs1})  ///
	cells(b(fmt(3) star) se(fmt(3) par)) ///
	note(Notes: The table reports intent-to-treat results from an OLS ///
		 regression of hourly electricity use on assignment to the treatment. ///
		 Columns 1 and 2 include all observations from May 15, 2015 to May ///
		 31, 2016. Columns 3 and 4 restrict the sample to the summer of 2015 ///
		 (May 15 to August 30). Columns 5 and 6 further limit the sample to ///
		 include only peak demand hours (3 PM to 8 PM). Standard errors are ///
		 clustered at the household. *, **, *** denote significance at the ///
		 10\%, 5\%, and 1\%level.) ///
	scalars("m_cont Mean Control Group Use" "weath Weather Controls" /// 
		    "hod Hour of Day FE" "date Calendar Date FE" ///
			 "hours Hours" "sam Sample") ///
	 mlabels((1) (2) (3) (4) (5) (6)) collabels(none) 
drop leuse1 leuse2	 
 
********************************************************************************
*TABLE A.4: WATER INTENT-TO-TREAT EFFECTS - LOG AND HYBERBOLIC SINE OF ELECTRICITY USE
********************************************************************************
gen lwuse1=log(wuse+1)
gen lwuse2=asinh(wuse)

************************************
*Log(euse+1)
eststo clear
*Column 1: Full Diff + weather + date + HOD FE
	eststo: reg lwuse1 ws temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1, vce(cluster HHID)
	sum wuse if e(sample) & ws==0
	scalar wuse1 = r(mean)
		estadd scalar m_cont = scalar(wuse1)
		estadd local weath "No"
		estadd local hod "No"	
		estadd local date "No"
		estadd local hours "All"
		estadd local sam "05/2015-05/2016"	
	est store A
	
*Column 2: Full Diff + weather + date + HOD FE + pre-treatment use
	eststo: reghdfe lwuse1 ws w_use_summer w_use_annual w_use_winter temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1, ///
			absorb(hour date) vce(cluster HHID)
	sum wuse if e(sample) & ws==0
	scalar wuse2 = r(mean)
		estadd scalar m_cont = scalar(wuse2)
		estadd local weath "Yes"
		estadd local hod "Yes"	
		estadd local date "Yes"
		estadd local hours "All"
		estadd local sam "05/2015-05/2016"	
	est store B
		
*Column 3: Summer 2015 Diff  + weather + date + HOD FE
	eststo: reg lwuse1 ws temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1 if  ym<=667, vce(cluster HHID)
	sum wuse if e(sample) & ws==0
	scalar wuse3 = r(mean)
		estadd scalar m_cont = scalar(wuse3)
		estadd local weath "No"
		estadd local hod "No"	
		estadd local date "No"
		estadd local hours "All"
		estadd local sam "05/2015-08/2015"					
	est store C
	
*Column 4: Summer 2015 Diff + weather + date + HOD FE + pre-treatment use
	eststo: reghdfe lwuse1 ws w_use_summer w_use_annual w_use_winter temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1 if ym<=667, ///
			absorb(hour date) vce(cluster HHID)
	sum wuse if e(sample) & ws==0
	scalar wuse4 = r(mean)
		estadd scalar m_cont = scalar(wuse4)
		estadd local weath "Yes"
		estadd local hod "Yes"	
		estadd local date "Yes"
		estadd local hours "All"
		estadd local sam "05/2015-08/2015"					
	est store D

*Column 5: Summer 2015 Diff Peak Hours + weather + date + HOD FE
	eststo: reg lwuse1 ws temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1 if ym<=667 & (hour>=15 & hour<=20), vce(cluster HHID)
	sum wuse if e(sample) & ws==0
	scalar wuse3 = r(mean)
		estadd scalar m_cont = scalar(wuse3)
		estadd local weath "No"
		estadd local hod "No"	
		estadd local date "No"
		estadd local hours "Peak"
		estadd local sam "05/2015-08/2015"					
	est store E
	
*Column 6: Summer 2015 Diff Peak Hours + weather + date + HOD FE + pre-treatment use
	eststo: reghdfe lwuse1 ws w_use_summer w_use_annual w_use_winter temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1 if ym<=667 & (hour>=15 & hour<=20), ///
			absorb(hour date) vce(cluster HHID)
	sum wuse if e(sample) & ws==0
	scalar wuse4 = r(mean)
		estadd scalar m_cont = scalar(wuse4)
		estadd local weath "Yes"
		estadd local hod "Yes"	
		estadd local date "Yes"
		estadd local hours "Peak"
		estadd local sam "05/2015-08/2015"					
	est store F
	
esttab A B C D E F using $tables\lwregs1_$outputdate.tex, replace label ///
    booktabs b(a2) nonumber ///
	drop(_cons temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1) ///
    starlevels(* 0.10 ** 0.05 *** 0.01) ///
	title(Water Intent-to-Treat Effects (Dependent Variable: ///
		  Water Use (gals/hr)) \label{lwregs1})  ///
	cells(b(fmt(3) star) se(fmt(3) par)) ///
	note(Notes: The table reports intent-to-treat results from an OLS ///
	     regression of hourly water use on assignment to the treatment. ///
		Columns 1 and 2 include all observations from May 15, 2015 to May ///
		31, 2016. Columns 3 and 4 restrict the sample to the summer of 2015 ///
		(May 15 to August 30). Columns 5 and 6 further limit the sample to ///
		include only peak demand hours (3 PM to 8 PM). Standard errors are ///
		clustered at the household. *, **, *** denote significance at the ///
		10\%, 5\%, and 1\%level.) ///
	scalars("m_cont Mean Control Group Use" "weath Weather Controls" /// 
		    "hod Hour of Day FE" "date Calendar Date FE" ///
			"hours Hours" "sam Sample") ///
	 mlabels((1) (2) (3) (4) (5) (6)) collabels(none) 		

	 
************************************
*sinh^-1(euse)	 	 
eststo clear
*Column 1: Full Diff + weather + date + HOD FE
	eststo: reg lwuse2 ws temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1, vce(cluster HHID)
	sum wuse if e(sample) & ws==0
	scalar wuse1 = r(mean)
		estadd scalar m_cont = scalar(wuse1)
		estadd local weath "No"
		estadd local hod "No"	
		estadd local date "No"
		estadd local hours "All"
		estadd local sam "05/2015-05/2016"	
	est store A
	
*Column 2: Full Diff + weather + date + HOD FE + pre-treatment use
	eststo: reghdfe lwuse2 ws w_use_summer w_use_annual w_use_winter temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1, ///
			absorb(hour date) vce(cluster HHID)
	sum wuse if e(sample) & ws==0
	scalar wuse2 = r(mean)
		estadd scalar m_cont = scalar(wuse2)
		estadd local weath "Yes"
		estadd local hod "Yes"	
		estadd local date "Yes"
		estadd local hours "All"
		estadd local sam "05/2015-05/2016"	
	est store B
		
*Column 3: Summer 2015 Diff  + weather + date + HOD FE
	eststo: reg lwuse2 ws temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1 if  ym<=667, vce(cluster HHID)
	sum wuse if e(sample) & ws==0
	scalar wuse3 = r(mean)
		estadd scalar m_cont = scalar(wuse3)
		estadd local weath "No"
		estadd local hod "No"	
		estadd local date "No"
		estadd local hours "All"
		estadd local sam "05/2015-08/2015"					
	est store C
	
*Column 4: Summer 2015 Diff + weather + date + HOD FE + pre-treatment use
	eststo: reghdfe lwuse2 ws w_use_summer w_use_annual w_use_winter temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1 if ym<=667, ///
			absorb(hour date) vce(cluster HHID)
	sum wuse if e(sample) & ws==0
	scalar wuse4 = r(mean)
		estadd scalar m_cont = scalar(wuse4)
		estadd local weath "Yes"
		estadd local hod "Yes"	
		estadd local date "Yes"
		estadd local hours "All"
		estadd local sam "05/2015-08/2015"					
	est store D

*Column 5: Summer 2015 Diff Peak Hours + weather + date + HOD FE
	eststo: reg lwuse2 ws temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1 if ym<=667 & (hour>=15 & hour<=20), vce(cluster HHID)
	sum wuse if e(sample) & ws==0
	scalar wuse3 = r(mean)
		estadd scalar m_cont = scalar(wuse3)
		estadd local weath "No"
		estadd local hod "No"	
		estadd local date "No"
		estadd local hours "Peak"
		estadd local sam "05/2015-08/2015"					
	est store E
	
*Column 6: Summer 2015 Diff Peak Hours + weather + date + HOD FE + pre-treatment use
	eststo: reghdfe lwuse2 ws w_use_summer w_use_annual w_use_winter temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1 if ym<=667 & (hour>=15 & hour<=20), ///
			absorb(hour date) vce(cluster HHID)
	sum wuse if e(sample) & ws==0
	scalar wuse4 = r(mean)
		estadd scalar m_cont = scalar(wuse4)
		estadd local weath "Yes"
		estadd local hod "Yes"	
		estadd local date "Yes"
		estadd local hours "Peak"
		estadd local sam "05/2015-08/2015"					
	est store F
	
esttab A B C D E F using $tables\lwregs2_$outputdate.tex, replace label ///
    booktabs b(a2) nonumber ///
	drop(_cons temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1) ///
    starlevels(* 0.10 ** 0.05 *** 0.01) ///
	title(Water Intent-to-Treat Effects (Dependent Variable: ///
		  Water Use (gals/hr)) \label{lwregs1})  ///
	cells(b(fmt(3) star) se(fmt(3) par)) ///
	note(Notes: The table reports intent-to-treat results from an OLS ///
	     regression of hourly water use on assignment to the treatment. ///
		Columns 1 and 2 include all observations from May 15, 2015 to May ///
		31, 2016. Columns 3 and 4 restrict the sample to the summer of 2015 ///
		(May 15 to August 30). Columns 5 and 6 further limit the sample to ///
		include only peak demand hours (3 PM to 8 PM). Standard errors are ///
		clustered at the household. *, **, *** denote significance at the ///
		10\%, 5\%, and 1\%level.) ///
	scalars("m_cont Mean Control Group Use" "weath Weather Controls" /// 
		    "hod Hour of Day FE" "date Calendar Date FE" ///
			"hours Hours" "sam Sample") ///
	 mlabels((1) (2) (3) (4) (5) (6)) collabels(none) 		
drop lwuse1 lwuse2
	 
	 
********************************************************************************
*FIGURE A1.A: ELECTRICITY INTENT TO TREAT EFFECTS OVER TIME - LOG ELECTRICITY USE
********************************************************************************

*Year-Month Indicators and Year-Month Interactions
tab ym, gen(ym_) 

 forval i = 1/13 {
	local j=663+`i'
	gen ym_ws_`i'=treat*ym_`i'
	label var ym_ws_`i' "WS, YM=`j'"
	disp `i'
}
*
gen leuse=log(euse+1)
compress

*Regression
reghdfe leuse ym_ws_1-ym_ws_13 ///
	  elec_use_summer elec_use_annual elec_use_winter ///
	  temp_60 temp_65 temp_70 temp_75 temp_80 precip_1, ///
	  absorb(hour date)  cluster(HHID)
			
*Saving results
gen ws_ym=.
	label var ws_ym "Month"
gen euse_ym=.
	gen euse_ym_ub=.
	gen euse_ym_lb=.
forval i = 1/13 {
	local j=663+`i'
	replace ws_ym=`j' in `i'
	replace euse_ym=_b[ym_ws_`i'] in `i'
	replace euse_ym_ub=(_b[ym_ws_`i']+1.96*_se[ym_ws_`i']) in `i'
	replace euse_ym_lb=(_b[ym_ws_`i']-1.96*_se[ym_ws_`i']) in `i'	
}
*		
*Regression
reghdfe leuse ym_ws_1-ym_ws_13 ///
	  elec_use_summer elec_use_annual elec_use_winter ///
	  temp_60 temp_65 temp_70 temp_75 temp_80 precip_1 if (hour>=15 & hour<=20), ///
	  absorb(hour date)  cluster(HHID)
gen euse_ym_peak=.
	gen euse_ym_peak_ub=.
	gen euse_ym_peak_lb=.
forval i = 1/13 {
	replace euse_ym_peak=_b[ym_ws_`i'] in `i'
	replace euse_ym_peak_ub=(_b[ym_ws_`i']+1.96*_se[ym_ws_`i']) in `i'
	replace euse_ym_peak_lb=(_b[ym_ws_`i']-1.96*_se[ym_ws_`i']) in `i'	
}
*	
preserve
keep ws_ym-euse_ym_peak_lb
drop if missing(ws_ym)

*Graph - TEs Over Time
format ws_ym %tm
twoway (line euse_ym_peak_ub ws_ym, lp(dash) lc(erose)) ///
		(line euse_ym_peak_lb ws_ym, lp(dash) lc(erose)) ///
       (connected euse_ym ws_ym, lc(edkblue) m(circle) mc(edkblue) msize(medium)  mfc(white) ) ///
	   (connected euse_ym_peak ws_ym, lc(cranberry) m(diamond) mc(cranberry) msize(medium)  mfc(white) ///
	   graphregion(color(white)) bgcolor(white)  ///
	   title("") xtitle("") ///  
	   ytit("Electricity ITT (kWh/hour)") ///
	   ylabel(-0.04(0.02)0.04, nogrid angle(hor)) ///
	   yline(0, lc(black) lp(solid)) ///
	   xlabel(664(4)676, angle(45)) ///
	   legend(order(3 "All Hours" 4 "Peak Hours") size(small) ///
	   region(lcolor(white)) cols(1) ring(0) position(5))  )
graph export $figs\leuse1_ym_controls_$outputdate.png, width(4000) replace	
restore
drop leuse 
drop ws_ym-euse_ym_peak_lb
********************************************************************************
*FIGURE A.1B: ELECTRICITY INTENT TO TREAT EFFECTS OVER TIME - ARCSIGN ELECTRICITY USE
********************************************************************************

gen leuse=asinh(euse)

*Regression
reghdfe leuse ym_ws_1-ym_ws_13 ///
	  elec_use_summer elec_use_annual elec_use_winter ///
	  temp_60 temp_65 temp_70 temp_75 temp_80 precip_1, ///
	  absorb(hour date)  cluster(HHID)
			
*Saving results
gen ws_ym=.
	label var ws_ym "Month"
gen euse_ym=.
	gen euse_ym_ub=.
	gen euse_ym_lb=.
forval i = 1/13 {
	local j=663+`i'
	replace ws_ym=`j' in `i'
	replace euse_ym=_b[ym_ws_`i'] in `i'
	replace euse_ym_ub=(_b[ym_ws_`i']+1.96*_se[ym_ws_`i']) in `i'
	replace euse_ym_lb=(_b[ym_ws_`i']-1.96*_se[ym_ws_`i']) in `i'	
}
*		
*Regression
reghdfe leuse ym_ws_1-ym_ws_13 ///
	  elec_use_summer elec_use_annual elec_use_winter ///
	  temp_60 temp_65 temp_70 temp_75 temp_80 precip_1 if (hour>=15 & hour<=20), ///
	  absorb(hour date)  cluster(HHID)
gen euse_ym_peak=.
	gen euse_ym_peak_ub=.
	gen euse_ym_peak_lb=.
forval i = 1/13 {
	replace euse_ym_peak=_b[ym_ws_`i'] in `i'
	replace euse_ym_peak_ub=(_b[ym_ws_`i']+1.96*_se[ym_ws_`i']) in `i'
	replace euse_ym_peak_lb=(_b[ym_ws_`i']-1.96*_se[ym_ws_`i']) in `i'	
}
*	
preserve
keep ws_ym-euse_ym_peak_lb
drop if missing(ws_ym)

*Graph - TEs Over Time
format ws_ym %tm
twoway (line euse_ym_peak_ub ws_ym, lp(dash) lc(erose)) ///
		(line euse_ym_peak_lb ws_ym, lp(dash) lc(erose)) ///
       (connected euse_ym ws_ym, lc(edkblue) m(circle) mc(edkblue) msize(medium)  mfc(white) ) ///
	   (connected euse_ym_peak ws_ym, lc(cranberry) m(diamond) mc(cranberry) msize(medium)  mfc(white) ///
	   graphregion(color(white)) bgcolor(white)  ///
	   title("") xtitle("") ///  
	   ytit("Electricity ITT (kWh/hour)") ///
	   ylabel(-0.04(0.02)0.04, nogrid angle(hor)) ///
	   yline(0, lc(black) lp(solid)) ///
	   xlabel(664(4)676, angle(45)) ///
	   legend(order(3 "All Hours" 4 "Peak Hours") size(small) ///
	   region(lcolor(white)) cols(1) ring(0) position(5))  )
graph export $figs\leuse2_ym_controls_$outputdate.png, width(4000) replace	
restore
drop leuse-euse_ym_peak_lb	 	 
drop ym_1-ym_ws_13
	
********************************************************************************
*FIGURE A.2A: ELECTRICITY HETEROGENEITY BY HOUR OF DAY: LOG ELECTRICITY USE
********************************************************************************
gen leuse=log(euse+1)

*Creating Hourly Indcators and Interactions 
forvalues i=0(1)23{
gen hod_`i'=0
	replace hod_`i'=1 if hour==`i'
	
gen hodxws_`i'=ws*hod_`i'
}
*
order hod_* hodxws_*, last

gen ws_hour=.
	label var ws_hour "Hour"
gen elec_h=.
	gen elec_h_ub=.
	gen elec_h_lb=.
gen elec_base_h=.

*Regression - Summer 2015 with Weather Controls and Date FEs
reghdfe leuse hod_1-hod_23 hodxws_* elec_use_summer elec_use_annual elec_use_winter temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1 if ym<=667, ///
	absorb(date) vce(cluster HHID)
	 
forvalues i=0(1)23{
	local j=`i'+1
	sum euse if e(sample) & ws==0 & hour==`i'	
		replace ws_hour=`i' in `j'
		replace elec_h=_b[hodxws_`i'] in `j'
		replace elec_h_ub=(_b[hodxws_`i']+1.96*_se[hodxws_`i']) in `j'
		replace elec_h_lb=(_b[hodxws_`i']-1.96*_se[hodxws_`i']) in `j'
		replace elec_base_h=r(mean) in `j'
	display `i'
}
*
*Graph - Hourly TEs & Avg. Use
twoway rcap elec_h_ub elec_h_lb ws_hour, lstyle(ci) lc(erose) || ///
       scatter elec_h ws_hour, m(diamond) mc(cranberry) msize(medium) /// 
	   graphregion(color(white)) bgcolor(white) ///
	   title("") xtitle("Hour of Day") ///  
	   ytit("Electricity ITT") ///
	   legend(off)  ///
	   xlabel(0(2)23) ///	   
	   yline(0, lcolor(black)) ///
	   ylabel(-0.02(0.01)0.01, nogrid)
graph export $figs\euse_log_hour_$outputdate.png, width(4000) replace		
drop leuse ws_hour-elec_base_h   
   
********************************************************************************
*FIGURE A.2b: ELECTRICITY HETEROGENEITY BY HOUR OF DAY: ARC-SINE ELECTRICITY USE
********************************************************************************
gen leuse=asinh(euse)

gen ws_hour=.
	label var ws_hour "Hour"
gen elec_h=.
	gen elec_h_ub=.
	gen elec_h_lb=.
gen elec_base_h=.

*Regression - Summer 2015 with Weather Controls and Date FEs
reghdfe leuse hod_1-hod_23 hodxws_* elec_use_summer elec_use_annual elec_use_winter temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1 if ym<=667, ///
	absorb(date) vce(cluster HHID)
	 
forvalues i=0(1)23{
	local j=`i'+1
	sum euse if e(sample) & ws==0 & hour==`i'	
		replace ws_hour=`i' in `j'
		replace elec_h=_b[hodxws_`i'] in `j'
		replace elec_h_ub=(_b[hodxws_`i']+1.96*_se[hodxws_`i']) in `j'
		replace elec_h_lb=(_b[hodxws_`i']-1.96*_se[hodxws_`i']) in `j'
		replace elec_base_h=r(mean) in `j'
	display `i'
}
*
*Graph - Hourly TEs & Avg. Use
twoway rcap elec_h_ub elec_h_lb ws_hour, lstyle(ci) lc(erose) || ///
       scatter elec_h ws_hour, m(diamond) mc(cranberry) msize(medium) /// 
	   graphregion(color(white)) bgcolor(white) ///
	   title("") xtitle("Hour of Day") ///  
	   ytit("Electricity ITT") ///
	   legend(off)  ///
	   xlabel(0(2)23) ///	   
	   yline(0, lcolor(black)) ///
	   ylabel(-0.02(0.01)0.01, nogrid)
graph export $figs\euse_arc_hour_$outputdate.png, width(4000) replace		  
drop leuse-elec_base_h
		  
		  
********************************************************************************
*FIGURE A.2C: WATER HETEROGENEITY BY HOUR OF DAY - LOG WATER USE
********************************************************************************
gen lwuse=log(wuse+1)
gen ws_hour=.
	label var ws_hour "Hour"
gen wtr_h=.
	gen wtr_h_ub=.
	gen wtr_h_lb=.
gen wtr_base_h=.

*Regression - Summer 2015 with Weather Controls and Date FEs
reghdfe lwuse hod_1-hod_23 hodxws_* w_use_summer w_use_annual w_use_winter temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1 if ym<=667, ///
	absorb(date) vce(cluster HHID)

forvalues i=0(1)23{
	local j=`i'+1
	sum wuse if e(sample) & ws==0 & hour==`i'	
		replace ws_hour=`i' in `j'
		replace wtr_h=_b[hodxws_`i'] in `j'
		replace wtr_h_ub=(_b[hodxws_`i']+1.96*_se[hodxws_`i']) in `j'
		replace wtr_h_lb=(_b[hodxws_`i']-1.96*_se[hodxws_`i']) in `j'
		replace wtr_base_h=r(mean) in `j'
	display `i'
}
*
*Graph - Hourly TEs & Avg. Use
twoway rcap wtr_h_ub wtr_h_lb ws_hour, lstyle(ci) lc(erose) || ///
       scatter wtr_h ws_hour, m(circle) mc(edkblue) msize(medium) /// 
	   graphregion(color(white)) bgcolor(white) ///
	   title("") xtitle("") ///  
	   ytit("Water ITT") ///
	   legend(off) ///
		   xlabel(0(2)23) ///	   
 xtick(0(2)23) ///	   
	yline(0, lcolor(black)) ///
	ylabel(-.1(0.05)0.05, nogrid)
graph export $figs\wuse_sum_hour_log_$outputdate.png, width(4000) replace
drop lwuse-wtr_base_h


********************************************************************************
*FIGURE A.2D: WATER HETEROGENEITY BY HOUR OF DAY - ARC-SINE WATER USE
********************************************************************************
gen lwuse=asinh(wuse)
gen ws_hour=.
	label var ws_hour "Hour"
gen wtr_h=.
	gen wtr_h_ub=.
	gen wtr_h_lb=.
gen wtr_base_h=.

*Regression - Summer 2015 with Weather Controls and Date FEs
reghdfe lwuse hod_1-hod_23 hodxws_* w_use_summer w_use_annual w_use_winter temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1 if ym<=667, ///
	absorb(date) vce(cluster HHID)

forvalues i=0(1)23{
	local j=`i'+1
	sum wuse if e(sample) & ws==0 & hour==`i'	
		replace ws_hour=`i' in `j'
		replace wtr_h=_b[hodxws_`i'] in `j'
		replace wtr_h_ub=(_b[hodxws_`i']+1.96*_se[hodxws_`i']) in `j'
		replace wtr_h_lb=(_b[hodxws_`i']-1.96*_se[hodxws_`i']) in `j'
		replace wtr_base_h=r(mean) in `j'
	display `i'
}
*
*Graph - Hourly TEs & Avg. Use
twoway rcap wtr_h_ub wtr_h_lb ws_hour, lstyle(ci) lc(erose) || ///
       scatter wtr_h ws_hour, m(circle) mc(edkblue) msize(medium) /// 
	   graphregion(color(white)) bgcolor(white) ///
	   title("") xtitle("") ///  
	   ytit("Water ITT") ///
	   legend(off) ///
		   xlabel(0(2)23) ///	   
 xtick(0(2)23) ///	   
	yline(0, lcolor(black)) ///
	ylabel(-.1(0.05)0.05, nogrid)
graph export $figs\wuse_sum_hour_arc_$outputdate.png, width(4000) replace
drop hod_0-hodxws_23
drop lwuse-wtr_base_h

	
********************************************************************************
*TABLE B.1: MULTIPLE HYPOTHESIS TESTING
********************************************************************************
frame copy robust mht
frame change mht

keep wuse euse ym ws hour HHID	   
gen euse_sum=euse if ym<=667 
gen euse_sum_peak=euse if ym<=667 & (hour>=15 & hour<=20)
gen wuse_sum=wuse if ym<=667 
gen wuse_sum_peak=wuse if ym<=667 & (hour>=15 & hour<=20)
gcollapse wuse euse euse_sum euse_sum_peak wuse_sum wuse_sum_peak ws, by(HHID)

mhtexp wuse wuse_sum wuse_sum_peak euse euse_sum euse_sum_peak, treatment(ws) bootstrap(5000)	
frame change robust
frame drop mht 

********************************************************************************
*TABLE D.2: ELECTRICITY INTENT TO TREAT EFFECT: NO POOLS
********************************************************************************
drop if pool==1

eststo clear
*Column 1: Full Diff + weather + date + HOD FE
	eststo: reghdfe euse ws temp_60 temp_65 temp_70 temp_75 temp_80 precip_1, ///
			absorb(hour date) vce(cluster HHID)
	sum euse if e(sample) & ws==0
	scalar euse1 = r(mean) 
		estadd scalar m_cont = scalar(euse1)
		estadd local weath "Yes"
		estadd local hod "Yes"	
		estadd local date "Yes"
		estadd local hours "All"
		estadd local sam "05/2015-05/2016"		
	est store A
	
*Column 2: Full Diff + weather + date + HOD FE + pre-treatment use
	eststo: reghdfe euse ws elec_use_summer elec_use_annual elec_use_winter temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1, ///
			absorb(hour date) vce(cluster HHID)
	sum euse if e(sample) & ws==0
	scalar euse2 = r(mean)
		estadd scalar m_cont = scalar(euse2)
		estadd local weath "Yes"
		estadd local hod "Yes"	
		estadd local date "Yes"
		estadd local hours "All"
		estadd local sam "05/2015-05/2016"	
	est store B
		
*Column 3: Summer 2015 Diff weather + date + HOD FE 
	eststo: reghdfe euse ws temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1 if ym<=667, ///
			absorb(hour date) vce(cluster HHID)
	sum euse if e(sample) & ws==0
	scalar euse3 = r(mean)
		estadd scalar m_cont = scalar(euse3)
		estadd local weath "Yes"
		estadd local hod "Yes"	
		estadd local date "Yes"
		estadd local hours "All"
		estadd local sam "05/2015-08/2015"					
	est store C
	
*Column 4: Summer 2015 Diff + weather + date + HOD FE + pre-treatment use
	eststo: reghdfe euse ws elec_use_summer elec_use_annual elec_use_winter temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1 if ym<=667, ///
			absorb(hour date) vce(cluster HHID)
	sum euse if e(sample) & ws==0
	scalar euse4 = r(mean)
		estadd scalar m_cont = scalar(euse4)
		estadd local weath "Yes"
		estadd local hod "Yes"	
		estadd local date "Yes"
		estadd local hours "All"
		estadd local sam "05/2015-08/2015"					
	est store D
		
*Column 5: Summer 2015 Diff weather + date + HOD FE - Peak Hours
	eststo: reghdfe euse ws temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1 if ym<=667 & (hour>=15 & hour<=20), ///
			absorb(hour date) vce(cluster HHID)
	sum euse if e(sample) & ws==0
	scalar euse5 = r(mean)
		estadd scalar m_cont = scalar(euse5)
		estadd local weath "Yes"
		estadd local hod "Yes"	
		estadd local date "Yes"
		estadd local hours "Peak"
		estadd local sam "05/2015-08/2015"					
	est store E
	
*Column 6: Summer 2015 Diff + weather + date + HOD FE  + pre-treatment use- Peak Hours
	eststo: reghdfe euse ws elec_use_summer elec_use_annual elec_use_winter temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1 if  ym<=667 & (hour>=15 & hour<=20), ///
			absorb(hour date) vce(cluster HHID)
	sum euse if e(sample) & ws==0
	scalar euse6 = r(mean)
		estadd scalar m_cont = scalar(euse6)
		estadd local weath "Yes"
		estadd local hod "Yes"	
		estadd local date "Yes"
		estadd local hours "Peak"
		estadd local sam "05/2015-08/2015"					
	est store F
	
esttab A B C D E F using $tables\eregsnopool_$outputdate.tex, replace label ///
    booktabs b(a2) nonumber ///
	drop(temp_60 temp_65 temp_70 temp_75 temp_80 temp_85 precip_1) ///
    starlevels(* 0.10 ** 0.05 *** 0.01) ///
	title(Electricity Intent-to-Treat Effects ///
	     (Dependent Variable: Electricity Use (kWh/hr)) \label{eregs_nopool})  ///
	cells(b(fmt(3) star) se(fmt(3) par)) ///
	note(Notes: The table reports intent-to-treat results from an OLS ///
		 regression of hourly electricity use on assignment to the treatment. ///
		 Columns 1 and 2 include all observations from May 15, 2015 to May ///
		 31, 2016. Columns 3 and 4 restrict the sample to the summer of 2015 ///
		 (May 15 to August 30). Columns 5 and 6 further limit the sample to ///
		 include only peak demand hours (3 PM to 8 PM). Standard errors are ///
		 clustered at the household. *, **, *** denote significance at the ///
		 10\%, 5\%, and 1\%level.) ///
	scalars("m_cont Mean Control Group Use" "weath Weather Controls" /// 
		    "hod Hour of Day FE" "date Calendar Date FE" ///
			 "hours Hours" "sam Sample") ///
	 mlabels((1) (2) (3) (4) (5) (6)) collabels(none) 	  

frame change data_main
frame drop robust 
	 

	